Been-Lon Chen, 2007. Notation: l stands for lambda. sig stands for sigma. gam stands for gamma. Del stands for capital delta. Om stands for capital omega.

```
> restart: with(plots):
```

Define various options to be used below

```
> alias(sigma=sig): alias(gamma=gam): alias(delta=delta_s):
> interface(displayprecision=4):
> axesfonts:=axesfont=[TIMES,ROMAN,6]:
> labelfonts:=labelfont=[TIMES,ROMAN,12]:
> webblue:=COLOR(RGB,.1,0,.55):
> webred:=COLOR(RGB,.9,.1,0):
> webgreen:=COLOR(RGB,.0,.5,.0):
> lightblue:=COLOR(RGB,0.196,0.6,0.8):
```

Parameter Assignments. The following are the parameter values requested by the referee in his 17 January comment.

```
> gam:=0.5: sig:=4.0: rho:=0.05: mu:=0.5: delta_s:=1: A:=0.1: B:=1.0: delta_k:=0:
> Del:=sig+(1-sig)*gam; Om:=1-mu+mu*gam;
```

Delta and Omega are defined in my comment.

$$Del := 2.5000$$

$$Om := 0.7500$$

Conditions on the Parameters.

```
> r:=A-delta_k;
```

$$r := 0.1000$$

```
> positive_growth_condition:=r-rho;
```

$$positive\_growth\_condition := 0.0500$$

```
> positive_asymptote_condition:=(sig-1)*(1-gam)*delta_s-rho;
```

If negative, Chen's Condition R is violated.

$$positive\_asymptote\_condition := 1.4500$$

Chen's Dynamic System (7a) and (7b). **These are Chen's equations as he writes them.**

```
> ldot:= diff(l(t),t) = l(t)*
  (A-delta_k+delta_s-gam*x(t)/l(t)-B*(1-(1-gam)*mu)*(x(t))^(mu)):
> xdot:= diff(x(t),t) = x(t)* ( (A-delta_k-rho+
  (gam*(1+(sig-1)/mu)*x(t)/l(t)-(rho+mu*delta_s+gam*(sig-1)*delta_s)+gam*mu*B*x(t)^(mu))
  *B*mu*l(t)*x(t)^(mu-1)+gam*(sig-1)*(B^(2)*mu*l(t)*x(t)^(2*mu-1)-delta_s)) /
  (sig+(sig+mu-1)*B*mu*l(t)*x(t)^(mu-1)) - (B*x(t)^mu-delta_s) ):
```

Chen's Dynamic System (7a) and (7b). **These are Chen's equations as I write them.**

```
> ldot:=diff(l(t),t) = l(t)* (r+delta_s-gam*x(t)/l(t)-B*(1-(1-gam)*mu)*(x(t))^(mu)):
> xdot:=diff(x(t),t) = x(t)*
  ((r-rho-(sig+(1-sig)*gam)*(B*(x(t))^(mu)-delta_s)-l(t)*(B*mu*(x(t))^(mu-1))*(rho+delta
  _s-gam*(x(t)/l(t))
  -B*(1-mu+mu*gam)*(x(t))^(mu)+(sig+(1-sig)*gam)*(B*(x(t))^(mu)-delta_s)))
  /(sig+(sig+mu-1)*l(t)*B*mu*(x(t))^(mu-1))):
```

Solving for a steady state (xss and lss indicate the BGP values, [0, xmax] and [0, lmax] indicate the range of the simulation)

```
> eql:=eval(ldot,{l(t)=l,x(t)=x}):
> eqx:=eval(xdot,{l(t)=l,x(t)=x}):
> systemss:=fsolve({eql,eqx},{l,x},x=0..10):
> lss:=subs(systemss,l):
> xss:=subs(systemss,x):
> xmin:=0: xmax:=2*xss: lmin:=0: lmax:=2.5*lss:
```

The range can be tweaked for better viewing effect.

Below I simulate the dynamic system. The maple command dfieldplot produces the phase diagram, while phaseportrait also allows to simulate particular trajectories.

```
> FNS:={x(t),l(t)}:
> SYS:={xdot,ldot}:
> phase:=DEtools[dfieldplot](SYS, FNS, t=0..100, scene=[x(t),l(t)], x=xmin..xmax,
  l=lmin..lmax, colour=lightblue, arrows=SLIM): display({phase}, axesfonts, labelfonts,
  title=`the Phase Diagram`);
```

the Phase Diagram



Below, I plot the phase diagram together with the one-dimensional stable manifold leading to the stationary state.

```
> INITS:=
  [[x(0)=0.999999*xss,l(0)=0.999999*lss],[x(0)=1.00000001*xss,l(0)=1.0000001*lss]]:
> p:=DEtools[phaseportrait](SYS, FNS, t=-100..0, INITS, scene=[x(t),l(t)], x=xmin..xmax,
  l=lmin..lmax, stepsize=0.5, colour=lightblue, arrows=slim, linestyle=1,
  linecolour=black, thickness=4, axesfonts, labelfonts): display(p, title=`The (unique)
  stable arm`);
```



There is a unique stationary state. The range of x and l can be extended: no multiple equilibria emerge. Below I stretch the range of x and l to beyond twice the value of the identified stationary state. Stretching x and l to the millions does not produce a second stationary state.

```
> DEtools[dfieldplot](SYS, FNS, t=0..100, scene=[x(t),l(t)], x=xmin..100, l=lmin..1000,
  colour=lightblue, arrows=slim,  title=`the Phase Diagram stretched out`);
```



Stretching the phase diagram inwards does not reveal a second stationary state either.

```
> DEtools[dfieldplot](SYS, FNS, t=0..100, scene=[x(t),l(t)], x=xmin..1, l=lmin..1,
  colour=lightblue, arrows=slim,  title=`the Phase Diagram squeazed in`);
```



At this point I have shown that in this numerical example there is a unique stationary state. Analytical proofs are given in my comment.

THE FOLLOWING IS EXTRA NUMERICAL ILLUSTRATION:

Below I graph the nullclines on top of the phase diagram. The nullclines are the lines associated with x'(t)=0, and l'(t)=0. Finding the nullclines requires solving the right-hand sides of equations (7a) and (7b).

```
> Lx:=x->-(r-rho-sig*B*x^mu+sig*delta_s-gam*B*x^mu+gam*delta_s+gam*sig*B*x^mu-gam*sig*delta_s+B*mu*x^(mu-1)*gam*x)/B/mu/(x^(mu-1))/(-B*mu*x^mu+B*x^mu*mu*gam-rho-delta_s+B*x^mu-gam*B*x^mu+gam*delta_s-sig*B*x^mu+sig*delta_s+gam*sig*B*x^mu-gam*sig*delta_s):
  Lx(x):
> Ll:=x->(gam*x)/(r+delta_s-B*(1-(1-gam)*mu)*x^(mu)): Ll(x):
> Den:=x->-B*mu*x^mu+B*x^mu*mu*gam-rho-delta_s+B*x^mu-gam*B*x^mu+gam*delta_s-sig*B*x^mu+sig*delta_s+gam*sig*B*x^mu-gam*sig*delta_s: Den(x):
> x_1:=fsolve(Den(x),x=xmin..xmax);
```

Undefined if Condition R is violated.

$$x\_1 := 0.6865$$

```
> x_1:=xmin:
```

WARNING: APPLY IF CONDITION R IS VIOLATED, SKIP OTHERWISE

```
> Num:=x->r-rho-sig*B*x^mu+sig*delta_s-gam*B*x^mu+gam*delta_s+gam*sig*B*x^mu-gam*sig*delta_s+B*mu*x^(mu-1)*gam*x: Num(x):
> x_2:=fsolve(Num(x),x=xmin..xmax):
```

No stationary state with positive value of l for x>x_2

```
> x_3:=((r+delta_s)/(B*Om))^(1/mu):
```

The nullcline l'(t)=0 has an asymptote for x=x_3.

```
> xmax:=x_3: lmin:=-2*lss: lmax:=2.5*lss:
> p1:=plot(Ll(x),x=xmin..0.999*xmax,thickness=2,color=webblue):
> p2:=plot(Lx(x),x=1.0001*x_1..xmax,thickness=2,color=webgreen):
> p3:=plot(Lx(x),x=0..0.999*x_1,numpoints=50000,thickness=2,color=webgreen):
> q1:=plot([x_1,l,l=lmin..lmax],thickness=2,linestyle=2,colour=webred):
> q2:=plot([x_2,l,l=lmin..lmax],thickness=2,linestyle=2,colour=webred):
```

```
> q3:=plot([x_3,l,l=lmin..lmax],thickness=2,linestyle=2,colour=webred):
```

Below is a graph of the nullclines. In green, the nullcline associated with x'(t)=0. In blue, the nullcline associated with l'(t)=0. No stationary state with positive l can exist to the right of x_2, marked by a vertical red line. The blue nullcline has an asymptote at x=x_3.

```
> display({p1,p2,p3,q1,q2,q3}, view=[xmin..xmax,lmin..lmax], title=`the nullclines`);
```




```
> display({p1,p2,p3}, view=[-0.01..0.1,-1..25], title=`the nullclines squeazed in`):
> display({phase,p,p1,p2,p3,q1},view=[xmin..xmax,lmin..lmax], axesfonts, labelfonts,
  title=`The phase diagram, the nullclines, and the stable arm`);
```



To make sure that the system above is indeed Chen's system (7a) and (7b), I compare the equations below with the ones in the text. Please check that these equations are correctly copied from Chen's equations (7a) and (7b).

REDEFINE LINES IN BLACK FOR PRINTING

```
> p1:=plot(Ll(x),x=xmin..0.999*xmax,thickness=2,color=black):
> p2:=plot(Lx(x),x=1.0001*x_1..xmax,thickness=2,color=black):
> p3:=plot(Lx(x),x=0..0.999*x_1,thickness=2,color=black):
> q1:=plot([x_1,l,l=lmin..lmax],thickness=2,linestyle=2,colour=black):
> q2:=plot([x_2,l,l=lmin..lmax],thickness=2,linestyle=2,colour=black):
> q3:=plot([x_3,l,l=lmin..lmax],thickness=2,linestyle=2,colour=black):
> xmax:=2*xss: lmin:=-2*lss: lmax:=2.5*lss:
> display({phase,p,p1,p2,p3,q1},view=[xmin..xmax,lmin..lmax]);
```




```
> plotsetup(ps,plotoutput=`Phase_Chen.ps`,plotoptions=`nocolor,portrait,noborder,axiswidth=300pt,axisheight=400pt`):
display({phase,p,p1,p2,p3,q1},view=[xmin..xmax,lmin..lmax],labels=[``,``],axesfonts,labelfonts); plotsetup(default):
>
```